

```

* 2 prisoners, light initially off

Room = enter . leave . Room ;

* suppose light initially off
LightOn = off . LightOff ;
LightOff = on . LightOn ;

* two prisoners: normal and counter
*
* the first time a prisoner finds the light off, he turns it on
P1 = 'enter . ('on . 'leave . P0 + 'off . 'on . 'leave . P1 ) ;
P0 = 'enter . 'leave . P0 ;
*
* when the counter finds the light on, he frees all
C1 = 'enter . ('off . 'free . 0 + 'on . 'off . 'leave . C1) ;

* system with 2 prisoners
S2 = (LightOff | Room | C1 | P1) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states: X = tau.X +
'free.0
* terminal state has configuration P0

*****

* 3 prisoners, light initially off

* when the counter finds the light on twice, he frees all
C2 = 'enter . ('off . 'leave . C1 + 'on . 'off . 'leave . C2) ;

* system with 3 prisoners
S3 = (LightOff | Room | C2 | P1 | P1) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states
* terminal state has configuration P0 | P0

*****

* initial status of the light unknown
Light = tau . LightOn + tau . LightOff ;

* 2 prisoners
S2' = (Light | Room | C1 | P1) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states
* terminal states has both configuration P0 and P1!

* 3 prisoners
S3' = (Light | Room | C2 | P1 | P1) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states
* terminal states has both configuration P0|P0 and P0|P1!

```

```

*****

* idea: count twice
P2 = 'enter . ('on . 'leave . P1 + 'off . 'on . 'leave . P2 ) ;

* 2 prisoners
S2 = (Light | Room | C2 | P2) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states X = tau.X +
'free.0
* terminal states has configuration P0 and P1

* 3 prisoners
C4 = 'enter . ('off . 'leave . C3 + 'on . 'off . 'leave . C4) ;
C3 = 'enter . ('off . 'leave . C2 + 'on . 'off . 'leave . C3) ;

S3 = (Light | Room | C4 | P2 | P2) \ {enter, leave, on, off} ;

* Explore: weak bisimulation collapse: two states X = tau.X +
'free.0
* terminal states has configuration P0 | P0 and P0 | P1

*****

```